Courtesy of Santos Bravo Yuste ☺
Dpt. Fisica. UEx. Badajoz. Spain
email: santos@unex.es
http://www1.unex.es/eweb/fisteor/santos/sby.html

An *explicit* finite difference method for fractional diffusion equations

Here it is provided a *Mathematica* notebook that uses the explicit difference numerical algorithm discussed in
Ref [1] = S. B. Yuste and L. Acedo, *On an explicit finite difference method for fractional diffusion equations*, SIAM Journal of Numerical Analysis, 52, 1862-74 (2005)

The fractional diffusion equation (FDE) we are going to solve is (see Eq. (1.2) of Ref[1]) :

$$\frac{\partial u}{\partial t} = K\, {}_oD^{1-\gamma}\, \frac{\partial^2 u}{\partial x^2}$$

In particular, we are going to consider two initial conditions:

□ (Case 1) Dirac' delta initial condition, u(x,0)= δ(x,0), and the boundary condition u(-∞,t)=u(∞,t)=0

□ (Case 2) u(x,0)=x(1-x) and Dirichlet boundary condition u(0,t)=u(1,t)=0

The solution u evaluated at position x_j at time t_m, $u(x_j,t_m)$, is denoted here as u[j,m,γ], where the third index γ is the anomalous diffusion exponent of the FDE.

The explicit difference scheme/algorithm for solving the FDE (1) [see Eq. (3.3) of Ref[1]) :

BDF1 Coefficients: w[k_, alfa_]≡$\omega_k^{(\alpha)}$ (see formula (2.6) of Ref[1]) :

```
w[0, alfa_] := 1;
w[k_, alfa_] := w[k, alfa] = (1 - (alfa + 1) / k) * w[k - 1, alfa]
```

In what follows,

lambda≡λ (weight factor; see Eq. (17) of [2]) ,

sfrac≡S (parameter defined in Eq. (19b) of [2]) and

Nx determines the space discretization of the interval in which the equation is going to be solve (the total number of points x_j is 2Nx+1)

uME[j, m, gamma] provides the value of the numeral solution at the point x_j at time t_m when the anomalous diffusion exponent γ is gamma.

```
uME[j_, m_, gamma_] := uME[j, m, gamma] = Module[{kk},
  kk = Sum[w[k, 1 - gamma] * {uME[j - 1, m - 1 - k, gamma],
      uME[j, m - 1 - k, gamma], uME[j + 1, m - 1 - k, gamma]}, {k, 0, m - 1}];
  uME[j, m - 1, gamma] + sfrac * (kk[[1]] - 2 * kk[[2]] + kk[[3]])]
```

Case 1:
Initial condition: Dirac' delta initial condition, u(x,0)= δ(x,0)

An example of case 1.

Parameters of the example: gamma=1/2, ht=0.0005,Kg=1, sfrac=0.33 (⇒the algorithm is stable), deltax=0.260307

(Notation: gamma=γ, ht=Δt , Kg=K, sfrac=S, deltax=Δx)

The exact solution (propagator) for $\gamma = 1/2$ and $K_\gamma = 1$:

```
uPropagador[x_, t_, 1 / 2] :=
 1/(4 t^(3/4)) (1/Gamma[3/4] 2 Sqrt[t] HypergeometricPFQ[{}, {1/2, 3/4}, -Abs[x]^4/(256 t)] +
   Abs[x] (-1/Sqrt[π] 2 t^(1/4) HypergeometricPFQ[{}, {3/4, 5/4}, -Abs[x]^4/(256 t)] +
     1/Gamma[1/4] Abs[x] HypergeometricPFQ[{}, {5/4, 3/2}, -Abs[x]^4/(256 t)]))
```

Here we clear the saved values of uME (the evaluation of this cell is required if you are going to work with several examples in a Mathematica session)

```
Clear[uME];
uME[j_, m_, gamma_] := uME[j, m, gamma] = Module[{kk},
  kk = Sum[w[k, 1 - gamma] * {uME[j - 1, m - 1 - k, gamma],
      uME[j, m - 1 - k, gamma], uME[j + 1, m - 1 - k, gamma]}, {k, 0, m - 1}];
  uME[j, m - 1, gamma] + sfrac * (kk[[1]] - 2 * kk[[2]] + kk[[3]])]
```

```
{gamma = 1 / 2, ht = 0.0005, Kg = 1.;
 sfrac = 0.33, deltax = (Kg * ht^gamma / sfrac) ^ (1 / 2)}
```

$$\left\{\frac{1}{2}, 0.0005, 0.33, 0.260307\right\}$$

2*Nx+1=number of spatial points

```
Nx = 25;
```

Parameters of the example

```
{sfracEfect = sfrac * (Sin[(2 * Nx - 1) * Pi / (4 * Nx)])^2, Sx = 2^(gamma - 2) // N}
```

```
{0.329674, 0.353553}
```

```
criterion = If[sfracEfect <= Sx, {"stable", " is not"}, {"unstable", " is "}];
Print["The algorithm is " <> criterion[[1]] <> " because sfracEfect=" <>
  ToString[sfracEfect] <> criterion[[2]] <> " larger than S$_x$=2^(gamma-2)=" <>
  ToString[Sx] <> " (see formula (3.12) of Ref[1])" ]
```

```
The algorithm is stable because sfracEfect=0.329674 is
  not larger than S$_x$=2^(gamma-2)=0.353553 (see formula (3.12) of Ref[1])
```

Here we set (we approximate) the initial condition [u(x,0)= δ(x,0)]

```
uME[nx_, 0, alfa_] := 1. / deltax /; nx == 0;
```

```
uME[nx_, 0, alfa_] := 0 /; nx ≠ 0;
```

```
Table[{nx, uME[nx, 0, 1]}, {nx, -Nx, Nx}]
```

```
{{-25, 0}, {-24, 0}, {-23, 0}, {-22, 0}, {-21, 0}, {-20, 0}, {-19, 0}, {-18, 0},
 {-17, 0}, {-16, 0}, {-15, 0}, {-14, 0}, {-13, 0}, {-12, 0}, {-11, 0}, {-10, 0},
 {-9, 0}, {-8, 0}, {-7, 0}, {-6, 0}, {-5, 0}, {-4, 0}, {-3, 0}, {-2, 0},
 {-1, 0}, {0, 3.84162}, {1, 0}, {2, 0}, {3, 0}, {4, 0}, {5, 0}, {6, 0}, {7, 0},
 {8, 0}, {9, 0}, {10, 0}, {11, 0}, {12, 0}, {13, 0}, {14, 0}, {15, 0}, {16, 0},
 {17, 0}, {18, 0}, {19, 0}, {20, 0}, {21, 0}, {22, 0}, {23, 0}, {24, 0}, {25, 0}}
```

Here we impose a Dirichlet boundary condition at the two ends of the interval (it is assumed that for the times considered the true solution satisfy, to a large extent, this boundary condition)

```
uL[t_] := 0.;
uR[t_] := 0;
```

```
uME[-Nx, m_, alfa_] := uL[m * ht];
uME[Nx, m_, alfa_] := uR[m * ht];
```

A small test to check that everything goes well:

```
{uME[1, 9, 0.5], uME[2, 4, 0.5]}
```

```
{0.857335, 0.459853}
```

soluME[t, gamma] is a function that generates a table with the solution u evaluated at time t. The table has the format {x, u (x, t)}

```
soluME[t_, gamma_] := Module[{mfin, kk, uMEfinalTab},
  mfin = Floor[t / ht];
  kk = Timing[Do[uMEfinalTab =
       Table[{j * deltax, uME[j, m, gamma]}, {j, -Nx, Nx}], {m, 0, mfin}]][[1]];
  Print["mfin=", mfin, ", t=", mfin * ht, ",  ht=", ht, ",  Delta x=",
   deltax, ",  S_γ=", sfrac, ",  Computation time=", kk,
   " :: Explicit method: Yuste-Acedo SIAM-JNA(2005)  "];
  uMEfinalTab]
```

Solution for t=0.005

```
tv = 0.005; soluME[tv, gamma]
```

```
mfin=10, t=0.005,  ht=0.0005,  Delta x=0.260307,   S_γ=0.33
 ,   Computation time=0.015625 :: Explicit method: Yuste-Acedo SIAM-JNA(2005)
```

```
{{-6.50767, 0.}, {-6.24736, 0.}, {-5.98706, 0.}, {-5.72675, 0.},
 {-5.46644, 0.}, {-5.20614, 0.}, {-4.94583, 0.}, {-4.68552, 0.}, {-4.42522, 0.},
 {-4.16491, 0.}, {-3.9046, 0.}, {-3.6443, 0.}, {-3.38399, 0.}, {-3.12368, 0.},
 {-2.86337, 0.}, {-2.60307, 0.0000588375}, {-2.34276, -0.000196125},
 {-2.08245, 0.00163221}, {-1.82215, -0.00281004}, {-1.56184, 0.016466},
 {-1.30153, -0.00985063}, {-1.04123, 0.0949135}, {-0.78092, 0.0451936},
 {-0.520614, 0.419054}, {-0.260307, 0.56257}, {0., 1.58756}, {0.260307, 0.56257},
 {0.520614, 0.419054}, {0.78092, 0.0451936}, {1.04123, 0.0949135},
 {1.30153, -0.00985063}, {1.56184, 0.016466}, {1.82215, -0.00281004},
 {2.08245, 0.00163221}, {2.34276, -0.000196125}, {2.60307, 0.0000588375},
 {2.86337, 0.}, {3.12368, 0.}, {3.38399, 0.}, {3.6443, 0.}, {3.9046, 0.},
 {4.16491, 0.}, {4.42522, 0.}, {4.68552, 0.}, {4.94583, 0.}, {5.20614, 0.},
 {5.46644, 0.}, {5.72675, 0.}, {5.98706, 0.}, {6.24736, 0.}, {6.50767, 0}}
```

Solution for t=0.1

```
tv = 0.1; soluME[tv, gamma]
```

```
mfin=200, t=0.1,  ht=0.0005,  Delta x=0.260307,   S_γ=0.33
 ,   Computation time=4.5625 :: Explicit method: Yuste-Acedo SIAM-JNA(2005)
```

$$\left\{\{-6.50767, 0.\}, \left\{-6.24736, 1.97272 \times 10^{-6}\right\}, \left\{-5.98706, 4.78243 \times 10^{-6}\right\},\right.$$
$$\left\{-5.72675, 9.58808 \times 10^{-6}\right\}, \{-5.46644, 0.0000182994\}, \{-5.20614, 0.0000342578\},$$
$$\{-4.94583, 0.0000633677\}, \{-4.68552, 0.000116018\}, \{-4.42522, 0.000210299\},$$
$$\{-4.16491, 0.000377341\}, \{-3.9046, 0.00067002\}, \{-3.6443, 0.00117693\},$$
$$\{-3.38399, 0.00204434\}, \{-3.12368, 0.00351\}, \{-2.86337, 0.00595396\},$$
$$\{-2.60307, 0.00997285\}, \{-2.34276, 0.016485\}, \{-2.08245, 0.0268733\},$$
$$\{-1.82215, 0.0431706\}, \{-1.56184, 0.0682826\}, \{-1.30153, 0.10623\},$$
$$\{-1.04123, 0.162364\}, \{-0.78092, 0.24346\}, \{-0.520614, 0.357541\},$$
$$\{-0.260307, 0.513197\}, \{0., 0.718078\}, \{0.260307, 0.513197\}, \{0.520614, 0.357541\},$$
$$\{0.78092, 0.24346\}, \{1.04123, 0.162364\}, \{1.30153, 0.10623\}, \{1.56184, 0.0682826\},$$
$$\{1.82215, 0.0431706\}, \{2.08245, 0.0268733\}, \{2.34276, 0.016485\}, \{2.60307, 0.00997285\},$$
$$\{2.86337, 0.00595396\}, \{3.12368, 0.00351\}, \{3.38399, 0.00204434\},$$
$$\{3.6443, 0.00117693\}, \{3.9046, 0.00067002\}, \{4.16491, 0.000377341\},$$
$$\{4.42522, 0.000210299\}, \{4.68552, 0.000116018\}, \{4.94583, 0.0000633677\},$$
$$\{5.20614, 0.0000342578\}, \{5.46644, 0.0000182994\}, \left\{5.72675, 9.58808 \times 10^{-6}\right\},$$
$$\left.\left\{5.98706, 4.78243 \times 10^{-6}\right\}, \left\{6.24736, 1.97272 \times 10^{-6}\right\}, \{6.50767, 0\}\right\}$$

Plot of the numerical solution for t=0.1

```
tv = 0.1;
figNum = ListPlot[soluME[tv, gamma],
  PlotRange → {{-Nx deltax, Nx deltax}, All}, PlotStyle → PointSize[0.02]]
```

```
mfin=200, t=0.1,  ht=0.0005,  Delta x=0.260307,   S_γ=0.33
 ,   Computation time=0. :: Explicit method: Yuste-Acedo SIAM-JNA(2005)
```



Plot of the exact solution for t=0.1

```
tv = 0.1;
figExact = Plot[uPropagador[x, tv, 1 / 2],
  {x, -6, 6}, PlotStyle → {Red, Thick}, PlotRange → All]
```



Comparison of the exact and numerical solution for t=0.1

```
Show[figNum, figExact, PlotRange → All]
```



Case 2:
Initial condition u(x,0)=x(1-x). Boundary condition: u(0,t)=u(1,t)=0.

An example of case 2.
Parameters of the example: gamma=1/2, ht=0.0005, Kg=1, sfrac=0.33 (⇒the algorithm is stable), deltax=0.260307
(Notation: gamma=γ, ht=Δt , Kg=K, sfrac=S, deltax=Δx)

Exact solution for $\gamma = 1/2$ and $K_\gamma = 1$

uExactx1mx[x,t,γ,Kg,*numterm*] is the exact solution of the FDE (1) written as a (finite) series with *numterm* terms (the exact solution corresponds to *numterm*→∞)

```
uExactx1mx[x_, t_, γ_, Kg_, numterm_] :=
 8 / Pi^3 * Sum[1 / (2 * n + 1)^3 * Sin[(2 * n + 1) * Pi * x] *
    MittagLefflerE[γ, -Kg * (2 * n + 1)^2 * Pi^2 * t^γ], {n, 0, numterm}]
```

(The Mittag-Leffler function for $\gamma = 1/2$ can be written in a simpler way: MittagLefflerE[1/2,z]=e^{z^2} Erfc[−z])

Here we clear the saved values of uME (the evaluation of this cell is required if you are going to work with several examples in a Mathematica session)

```
Clear[uME];
uME[j_, m_, gamma_] := uME[j, m, gamma] = Module[{kk},
   kk = Sum[w[k, 1 - gamma] * {uME[j - 1, m - 1 - k, gamma],
       uME[j, m - 1 - k, gamma], uME[j + 1, m - 1 - k, gamma]}, {k, 0, m - 1}];
   uME[j, m - 1, gamma] + sfrac * (kk[[1]] - 2 * kk[[2]] + kk[[3]])]
```

2*Nx+1=number of spatial points

```
Nx = 10;
```

Parameters of the example

```
{gamma = 0.5, deltax = 1. / (2 * Nx), Kg = 1.;
 sfrac = .33, ht = (sfrac * deltax^2 / Kg) ^ (1 / gamma)}
```

```
{0.5, 0.05, 0.33, 6.80625 × 10^-7}
```

```
{ sfracEfect = sfrac * (Sin[ (2 * Nx - 1) * Pi / (4 * Nx) ]) ^2, Sx = 2^ (gamma - 2) // N}
```

```
{0.327969, 0.353553}
```

```
criterion = If [sfracEfect <= Sx, {"stable", " is not"}, {"unstable", " is "}];
Print["The algorithm is " <> criterion[[1]] <> " because sfracEfect=" <>
  ToString[sfracEfect] <> criterion[[2]] <> " larger than Sₓ=2^(gamma-2)=" <>
  ToString[Sx] <> " (see formula (3.12) of Ref[1])" ]
```

```
The algorithm is stable because sfracEfect=0.327969 is
  not larger than Sₓ=2^(gamma-2)=0.353553 (see formula (3.12) of Ref[1])
```

Initial condition : u(x,0)=x(1-x)

```
uME[nx_, 0, alfa_] := (nx + Nx) * deltax * (1 - (nx + Nx) * deltax)
```

```
Table[{nx, uME[nx, 0, 1]}, {nx, -Nx, Nx}] // N
```

```
{{-10., 0.}, {-9., 0.0475}, {-8., 0.09}, {-7., 0.1275}, {-6., 0.16},
 {-5., 0.1875}, {-4., 0.21}, {-3., 0.2275}, {-2., 0.24}, {-1., 0.2475},
 {0., 0.25}, {1., 0.2475}, {2., 0.24}, {3., 0.2275}, {4., 0.21},
 {5., 0.1875}, {6., 0.16}, {7., 0.1275}, {8., 0.09}, {9., 0.0475}, {10., 0.}}
```

Boundary condition: u(0,t)=u(1,t)=0

```
uL[t_] := 0.;
uR[t_] := 0;
```

```
uME[-Nx, m_, alfa_] := uL[m * ht];
uME[Nx, m_, alfa_] := uR[m * ht];
```

soluME[t, gamma] is a function that generates a table with the solution u evaluated at time t. The table has the format {x, u (x, t)}

```
soluME[t_, gamma_] := Module[{mfin, kk, uMEfinalTab, eps = $MachineEpsilon},
  mfin = Floor[(t + eps) / ht];
  kk = Timing[Do[uMEfinalTab = Table[{(j + Nx) * deltax // N, uME[j, m, gamma]},
        {j, -Nx, Nx}], {m, 0, mfin}]][[1]];
  Print["mfin=", mfin, " : t=", mfin * ht, " : ht=", ht, " : Delta x=",
   deltax, " : S_γ=", sfrac, " : Computation time=", kk,
   " : Explicit method: Yuste-Acedo SIAM-JNA(2005)   "];
  uMEfinalTab]
```

```
mfin = 8; soluME[mfin * ht, gamma]
```

```
mfin=8 : t=5.445×10^-6 : ht=6.80625×10^-7 : Delta x=0.05 : S_γ=0.33
  : Computation time=0. : Explicit method: Yuste-Acedo SIAM-JNA(2005)
```

```
{{0., 0.}, {0.05, 0.0441515}, {0.1, 0.0854012}, {0.15, 0.1225},
 {0.2, 0.154857}, {0.25, 0.18233}, {0.3, 0.204817}, {0.35, 0.222316},
 {0.4, 0.234816}, {0.45, 0.242316}, {0.5, 0.244816}, {0.55, 0.242316},
 {0.6, 0.234816}, {0.65, 0.222316}, {0.7, 0.204817}, {0.75, 0.18233},
 {0.8, 0.154857}, {0.85, 0.1225}, {0.9, 0.0854012}, {0.95, 0.0441515}, {1., 0}}
```

Plot of the numerical solution for t=200*Δt=0.000136125

```
mfin = 200;
figNum =
 ListPlot[soluME[mfin * ht, gamma], PlotRange → All, PlotStyle → PointSize[0.02]]
```

```
mfin=200 : t=0.000136125 : ht=6.80625×10^-7 : Delta x=0.05 : S_γ=0.33
  : Computation time=1.6875 : Explicit method: Yuste-Acedo SIAM-JNA(2005)
```



Plot of the exact solution for t=200*Δt=0.000136125

```
tv = 200 * ht;
figExact = Plot[uExactx1mx[x, tv, 1 / 2, 1, 20],
   {x, 0, 1}, PlotStyle → {Red, Thick}, PlotRange → All]
```



Comparison of the exact and numerical solution for t=200*Δt=0.000136125

```
Show[figNum, figExact, PlotRange → All]
```